

ECGR 6118  
 Computer Project: Segmentation  
 Student Name: \_\_\_\_\_


(copyright T. Weldon, 2006)

**For this project, you may use mathcad or NetBeans  
 However, MathCad will provide the most direct solution**

Complete the tasks below and turn in a project report  
 answering all questions.

```
infile := READBMP("grass1weave55t1.gif")

rr := 0..rows(infile) - 1   cc := 0..cols(infile) - 1
rows(infile) = 256         cols(infile) = 256
max(infile) = 251          min(infile) = 61
```



**Define 2 Gabor filters in frequency domain**

```
ri := rows(infile)   ci := cols(infile)

r0 := 50   c0 := 50   sigma := 20
```

$$gab1_{rr,cc} := e^{-\frac{\left[ \left( \text{if} \left( rr < \frac{ri}{2}, rr, rr-ri \right) - r0 \right)^2 + \left( \text{if} \left( cc < \frac{ci}{2}, cc, cc-ci \right) - c0 \right)^2 \right]}{\sigma^2}}$$

```
r02 := 15   c02 := 0   sigma2 := 10
```

$$gab2_{rr,cc} := e^{-\frac{\left[ \left( \text{if} \left( rr < \frac{ri}{2}, rr, rr-ri \right) - r02 \right)^2 + \left( \text{if} \left( cc < \frac{ci}{2}, cc, cc-ci \right) - c02 \right)^2 \right]}{\sigma2^2}}$$

1. Why is the "if(rr<ri/2,rr,rr-ri)" needed in the definition of the above Gabor filters in the frequency domain? (Note: gab1 and gab2 are frequency responses of a Gabor filter)  
 Hint: try setting r0=0 and plot gab1 with and without the "if"

**Filter the image (AC component only) using both filters**

$$X := \text{cfft}(\text{infile}) \quad x_{ac} := \text{infile} - \text{mean}(\text{infile}) \quad X_{ac} := \text{cfft}(x_{ac})$$

$$Y1 := \overrightarrow{(\text{gab1} \cdot X_{ac})}$$

$$Y2 := \overrightarrow{(\text{gab2} \cdot X_{ac})}$$

$$\log Y1 := \overrightarrow{50 \log(|Y1| + 0.00000001)}$$

$$\log Y2 := \overrightarrow{50 \log(|Y2| + 0.00000001)}$$



$$\max Y1 := \max(\log Y1) \quad \max Y1 = 89.64$$

$$\max Y2 := \max(\log Y2) \quad \max Y2 = 123.31$$

$$\log Y1 := \log Y1 - \max Y1 + 250$$

$$\log Y2 := \log Y2 - \max Y2 + 250$$

$$\log Y1_{rr,cc} := \text{if}(\log Y1_{rr,cc} < 0, 0, \text{floor}(\log Y1_{rr,cc})) \quad \log Y2_{rr,cc} := \text{if}(\log Y2_{rr,cc} < 0, 0, \text{floor}(\log Y2_{rr,cc}))$$



**2.** Why does the Gabor filter frequency response appear in all four corners of the right-hand plot?

**Take inverse transform to recover filtered images**

$$ygab1 := \text{icfft}(Y1)$$

$$ygab1 := \overrightarrow{|ygab1|}$$

$$\text{miny1} := \min(ygab1) \quad \text{maxy1} := \max(ygab1)$$



$$\text{miny1} = 5.902 \times 10^{-6} \quad \text{maxy1} = 9.499$$

$$ygab2 := \text{icfft}(Y2)$$

$$ygab2 := \overrightarrow{|ygab2|}$$

$$\text{miny2} := \min(ygab2) \quad \text{maxy2} := \max(ygab2)$$

$$\text{miny2} = 0.012 \quad \text{maxy2} = 19.507$$



3. Explain which of the two filter-output features above appears to be the better feature?

**To train classifier, obtain sample data for first texture and both filters**

```
infile1 := READBMP("grass1t1.gif")
```


```
rr := 0..rows(infile1) - 1 cc := 0..cols(infile1) - 1 rows(infile1) = 256 cols(infile1) = 256
```

```
max(infile1) = 251 min(infile1) = 67
```

```
rr := 0..rows(infile1) - 1 cc := 0..cols(infile1) - 1
```

```
X1 := cfft(infile1)
```

```
xac1 := infile1 - mean(infile1) Xac1 := cfft(xac1)
```



$$Ya1 := \overrightarrow{(gab1 \cdot Xac1)}$$

$$Ya2 := \overrightarrow{(gab2 \cdot Xac1)}$$

```
yagab1 := icfft(Ya1)
```



```
yagab2 := icfft(Ya2)
```

$$yagab1 := |\overrightarrow{yagab1}|$$

$$yagab2 := |\overrightarrow{yagab2}|$$

```
minya1 := min(yagab1) maxya1 := max(yagab1) minya2 := min(yagab2) maxya2 := max(yagab2)
```

```
minya1 = 0.012 maxya1 = 9.501 minya2 = 0.026 maxya2 = 19.159
```



**4.** Why does the left filter-output feature above appear to be less coarse than the one on the right (the left has finer resolution features)?

### Measure statistics of filtered samples for Bayesian Multivariate Gaussian classifier

$$oo_0 := \text{yagab1} \quad oo_1 := \text{yagab2}$$

$$uu1 := \begin{pmatrix} \text{mean}(oo_0) \\ \text{mean}(oo_1) \end{pmatrix} \quad uu1 = \begin{pmatrix} 2.181 \\ 5.454 \end{pmatrix}$$

$$ssr := 0.. \text{rows}(uu1) - 1 \quad ssc := ssr$$

$$RR1_{ssr, ssc} := \text{mean}(\overrightarrow{oo_{ssr} oo_{ssc}}) \quad \Sigma\Sigma1 := RR1 - uu1 \cdot uu1^T$$

$$RR1 = \begin{pmatrix} 6.462 & 12.084 \\ 12.084 & 38.278 \end{pmatrix} \quad \Sigma\Sigma1 = \begin{pmatrix} 1.705 & 0.188 \\ 0.188 & 8.528 \end{pmatrix}$$

### Gather exemplars of filtered samples for use in nearest-neighbor classifier

$$knn := 10 \quad nn := 0.. knn - 1$$

$$\text{neighborsA}_{nn} := \begin{pmatrix} \text{yagab1}_{10 \cdot nn + 10, 10 \cdot nn + 10} \\ \text{yagab2}_{10 \cdot nn + 10, 10 \cdot nn + 10} \end{pmatrix} \quad \text{neighborsA}_0 = \begin{pmatrix} 1.607 \\ 6.227 \end{pmatrix}$$

**Obtain sample data for second texture and both filters**

`infile2 := READBMP("weave55t1.gif")`


`rr := 0..rows(infile2) - 1 cc := 0..cols(infile2) - 1 rows(infile2) = 256 cols(infile2) = 256`

`max(infile2) = 228 min(infile2) = 66`

`rr := 0..rows(infile2) - 1 cc := 0..cols(infile2) - 1`

`X1 := cfft(infile2)`

`xac2 := infile2 - mean(infile2) Xac2 := cfft(xac2)`



$$Yb1 := \overrightarrow{(gab1 \cdot Xac2)}$$

$$Yb2 := \overrightarrow{(gab2 \cdot Xac2)}$$

$$ybgab1 := \text{icfft}(Yb1)$$



$$ybgab2 := \text{icfft}(Yb2)$$

$$ybgab1 := |ybgab1|$$

$$ybgab2 := |ybgab2|$$

`minyb1 := min(ybgab1) maxyb1 := max(ybgab1) minyb2 := min(ybgab2) maxyb2 := max(ybgab2)`

`minyb1 =  $2.677 \times 10^{-3}$  maxyb1 = 3.047 minyb2 = 0.017 maxyb2 = 27.146`



### Measure statistics of filtered samples for Bayesian Multivariate Gaussian classifier

$$oo_0 := ybgab1 \quad oo_1 := ybgab2$$

$$uu2 := \begin{pmatrix} \text{mean}(oo_0) \\ \text{mean}(oo_1) \end{pmatrix} \quad uu2 = \begin{pmatrix} 0.642 \\ 7.464 \end{pmatrix}$$

$$ssr := 0.. \text{rows}(uu1) - 1 \quad ssc := ssr$$

$$RR2_{ssr, ssc} := \text{mean} \left( \overrightarrow{oo_{ssr} oo_{ssc}} \right) \quad \Sigma\Sigma2 := RR2 - uu2 \cdot uu2^T$$

$$RR1 = \begin{pmatrix} 6.462 & 12.084 \\ 12.084 & 38.278 \end{pmatrix} \quad \Sigma\Sigma1 = \begin{pmatrix} 1.705 & 0.188 \\ 0.188 & 8.528 \end{pmatrix}$$

### Gather exemplars of filtered samples for use in nearest-neighbor classifier

$$knn := 10 \quad nn := 0.. knn - 1$$

$$\text{neighbors}_{B_{nn}} := \begin{pmatrix} ybgab1_{10 \cdot nn + 10, 10 \cdot nn + 10} \\ ybgab2_{10 \cdot nn + 10, 10 \cdot nn + 10} \end{pmatrix} \quad \text{neighbors}_{B_0} = \begin{pmatrix} 0.335 \\ 7.462 \end{pmatrix}$$

B-distance

$$B(u1, u2, s1, s2) := \left| \frac{1}{8} \cdot ((u1 - u2))^T \cdot \left( \frac{s1 + s2}{2} \right)^{-1} \cdot (u1 - u2) + \frac{1}{2} \cdot \ln \left[ \frac{\left| \frac{(s1 + s2)}{2} \right|}{\sqrt{|s1|} \cdot \sqrt{|s2|}} \right] \right|$$

$$B(uu1, uu2, \Sigma\Sigma1, \Sigma\Sigma2) = 0.706 \quad uu1 = \begin{pmatrix} 2.181 \\ 5.454 \end{pmatrix} \quad \Sigma\Sigma1 = \begin{pmatrix} 1.705 & 0.188 \\ 0.188 & 8.528 \end{pmatrix}$$

$$\frac{1}{2} e^{-B(uu1, uu2, \Sigma\Sigma1, \Sigma\Sigma2)} = 0.247 \quad uu2 = \begin{pmatrix} 0.642 \\ 7.464 \end{pmatrix} \quad \Sigma\Sigma2 = \begin{pmatrix} 0.141 & -0.108 \\ -0.108 & 15.974 \end{pmatrix}$$

### Compute probabilities based on multivariate Gaussian

mvgauss computes Gaussian pdf for vector  $x=[i_0, i_1 \dots]^T$  is comprised of feature images  $i_0, i_1$ , etc, and given a Gaussian pdf with mean vector =  $u$ , and covariance matrix  $s$

```

mvgauss(x, u, s) :=
  N ← rows(x)
  nn ← 0..N - 1
  dets ← |s|
  c1 ← ln  $\left[ \frac{1}{\frac{N}{(2 \cdot \pi)^2 \cdot \sqrt{dets}}} \right]$ 
  sinv ← s-1
  for rr ∈ 0..rows(x0) - 1
    for cc ∈ 0..cols(x0) - 1
      for nn ∈ 0..rows(x) - 1
        u1nn ← (xnn)rr, cc
        temp ← c1 - 0.5 · [(u1 - u)T · sinv · (u1 - u)]
        yrr, cc ← temp0
        yrr, cc ← eyrr, cc
  y

```

generate the vector of feature images, fv

$fv_0 := ygab1$      $fv_1 := ygab2$



### Create probability images based on multivariate Gaussian

$p1img := mvgauss(fv, uu1, \Sigma1)$

$maxp1img := \max(p1img)$

$maxp1img = 0.042$



$\min(p1img) = 4.677 \times 10^{-9}$

$p2img := mvgauss(fv, uu2, \Sigma2)$

$maxp2img := \max(p2img)$

$maxp2img = 0.106$



$\min(p2img) = 0$



5. Explain what the lighter and darker regions mean in the right-hand image above?

Create segmented image based on multivariate Gaussian

$$\text{segimg}_{rr,cc} := \text{if}(p1\text{img}_{rr,cc} > p2\text{img}_{rr,cc}, 0, 250)$$



6. Find a way to measure the error on the segmented images, and report the measured error for the above Bayesian multivariate Gaussian segmentation.

## Nearest-neighbor classifier

nearestneighbor computes the nearest neighbor,  
where vector  $fv=[i_0, i_1 \dots]^T$  is comprised of feature images  $i_0, i_1$ , etc,  
where vector  $ndat=[na, nb \dots]^T$  is comprised of vectors  $na, nb$ , etc,  
where vectors  $na, nb, nc$ , correspond to classes  $a, b, c$ , etc  
where  $na=[a_1, a_2 \dots]^T$  where each element  $a_1, a_2$  is a sample feature vector for class  $a$ ,  
where  $nb=[b_1, b_2 \dots]^T$  where each element  $b_1, b_2$  is a sample feature vector for class  $b$ ,



```
nearestneighbor(ndat, fv) :=
  nclass ← rows(ndat)
  nfeat ← rows(fv)
  nnei ← rows(ndat0)
  for rr ∈ 0..rows(fv0) - 1
    for cc ∈ 0..cols(fv0) - 1
      for nf ∈ 0..nfeat - 1
        vvnf ← (fvnf)rr,cc
        class ← 0
        dmin ← |vv - [(ndat0)0]|
        for nc ∈ 0..nclass - 1
          for nn ∈ 0..nnei - 1
            d ← |vv - [(ndatnc)nn]|
            class ← nc if d < dmin
            dmin ← d if d < dmin
          yrr,cc ← class
```

y

## Create segmented image based on nearest-neighbor classifier

```
ndat0 := neighborsA      ndat1 := neighborsB
```

```
nearneigh := 250·nearestneighbor(ndat, fv)
```



7. Find a way to measure the error on the segmented images, and report the measured error for the above nearest-neighbor segmentation.